Pensieve header: Programs for β-calculus, development notebook.

KnotTheory

```
<< KnotTheory`
```

KnotTheory

```
Loading KnotTheory` version of February 5, 2013, 3:48:46.4762.
Read more at http://katlas.org/wiki/KnotTheory.
```

Initialization

```
βSimp = Factor; SetAttributes[βCollect, Listable];
βCollect[B[ω_, Λ_]] := B[βSimp[ω],
   Collect[Λ, h_, Collect[#, t_, βSimp] &]];
βForm[B[ω_, Λ_]] := Module[{ts, hs, M},
   ts = Union[Cases[B[ω, Λ], (t | T)_s_ ⧴ s, Infinity]];
   hs = Union[Cases[B[ω, Λ], h_s_ ⧴ s, Infinity]];
   M = Outer[βSimp[Coefficient[Λ, h_#1 t_#2]] &, hs, ts];
   PrependTo[M, t_# & /@ ts];
   M = Prepend[Transpose[M], Prepend[h_# & /@ hs, ω]];
   MatrixForm[M]];
βForm[else_] := else /. β_B ⧴ βForm[β];
Format[β_B, StandardForm] := βForm[β];
```

Program

```
⟨μ_⟩ := μ /. t_ → 1;
tm_x_,y_→z_[β_] := βCollect[β /. {t_x|y → t_z, T_x|y → T_z}];
hm_x_,y_→z_[B[ω_, Λ_]] := Module[
   {α = D[Λ, h_x], β = D[Λ, h_y], γ = Λ /. h_x|y → 0},
   B[ω, (α + (1 + ⟨α⟩) β) h_z + γ] // βCollect];
sw_x_,y_[B[ω_, Λ_]] := Module[{α, β, γ, δ, ϵ},
   α = Coefficient[Λ, h_y t_x]; β = D[Λ, t_x] /. h_y → 0;
   γ = D[Λ, h_y] /. t_x → 0;    δ = Λ /. h_y | t_x → 0;
   ϵ = 1 + α;
   B[ω * ϵ, α (1 + ⟨γ⟩ / ϵ) h_y t_x + β (1 + ⟨γ⟩ / ϵ) t_x
         + γ / ϵ h_y              + δ - γ * β / ϵ
    ] // βCollect];
gm_x_,y_→z_[β_] := β // sw_xy // hm_xy→z // tm_xy→z;
B /: B[ω1_, Λ1_] B[ω2_, Λ2_] := B[ω1 * ω2, Λ1 + Λ2];
(R⁺)_x_,y_ := B[1, (T_x - 1) t_x h_y];
(R⁻)_x_,y_ := B[1, ((T_x)^-1 - 1) t_x h_y];
```

tm

$$\{\beta = \mathbf{B}[\omega, \mathbf{Sum}[\alpha_{2\,i+j-6}\, t_i\, h_j,\ \{i, 1, 4\},\ \{j, 5, 6\}]],$$
$$O_1 = \beta\ //\ tm_{12\to 1}\ //\ tm_{13\to 1},$$
$$O_2 = \beta\ //\ tm_{23\to 2}\ //\ tm_{12\to 1},$$
$$O_1 == O_2$$
$$\}\ //\ \mathbf{ColumnForm}$$

tm

$$\begin{pmatrix} \omega & h_5 & h_6 \\ t_1 & \alpha_1 & \alpha_2 \\ t_2 & \alpha_3 & \alpha_4 \\ t_3 & \alpha_5 & \alpha_6 \\ t_4 & \alpha_7 & \alpha_8 \end{pmatrix}$$
$$\begin{pmatrix} \omega & h_5 & h_6 \\ t_1 & \alpha_1 + \alpha_3 + \alpha_5 & \alpha_2 + \alpha_4 + \alpha_6 \\ t_4 & \alpha_7 & \alpha_8 \end{pmatrix}$$
$$\begin{pmatrix} \omega & h_5 & h_6 \\ t_1 & \alpha_1 + \alpha_3 + \alpha_5 & \alpha_2 + \alpha_4 + \alpha_6 \\ t_4 & \alpha_7 & \alpha_8 \end{pmatrix}$$
True

hm

$$\{\beta = \mathbf{B}[\omega, \mathbf{Sum}[\alpha_{4\,i+j-6}\, t_i\, h_j,\ \{i, 1, 2\},\ \{j, 3, 6\}]],$$
$$O_1 = \beta\ //\ hm_{34\to 3}\ //\ hm_{35\to 3},$$
$$O_2 = \beta\ //\ hm_{45\to 4}\ //\ hm_{34\to 3};$$
$$O_1 == O_2$$
$$\}\ /.\ \alpha_{i\_} :\to \hat{i}\ //\ \mathbf{ColumnForm}$$

hm

$$\begin{pmatrix} \omega & h_3 & h_4 & h_5 & h_6 \\ t_1 & \hat{1} & \hat{2} & \hat{3} & \hat{4} \\ t_2 & \hat{5} & \hat{6} & \hat{7} & \hat{8} \end{pmatrix}$$
$$\begin{pmatrix} \omega & h_3 & h_6 \\ t_1 & \hat{1} + \hat{2} + \hat{1}\,\hat{2} + \hat{3} + \hat{1}\,\hat{3} + \hat{2}\,\hat{3} + \hat{1}\,\hat{2}\,\hat{3} + \hat{2}\,\hat{5} + \hat{3}\,\hat{5} + \hat{2}\,\hat{3}\,\hat{5} + \hat{3}\,\hat{6} + \hat{1}\,\hat{3}\,\hat{6} + \hat{3}\,\hat{5}\,\hat{6} & \hat{4} \\ t_2 & \hat{5} + \hat{6} + \hat{1}\,\hat{6} + \hat{5}\,\hat{6} + \hat{7} + \hat{1}\,\hat{7} + \hat{2}\,\hat{7} + \hat{1}\,\hat{2}\,\hat{7} + \hat{5}\,\hat{7} + \hat{2}\,\hat{5}\,\hat{7} + \hat{6}\,\hat{7} + \hat{1}\,\hat{6}\,\hat{7} + \hat{5}\,\hat{6}\,\hat{7} & \hat{8} \end{pmatrix}$$
True

htt

$$\{\beta = \mathbf{B}[\omega, \mathbf{Sum}[\alpha_{2\,i+j-5}\, t_i\, h_j,\ \{i, 1, 3\},\ \{j, 4, 5\}]],$$
$$O_1 = \beta\ //\ tm_{12\to 1}\ //\ sw_{14},$$
$$O_2 = \beta\ //\ sw_{24}\ //\ sw_{14}\ //\ tm_{12\to 1};$$
$$O_1 == O_2\}$$

htt

$$\left\{\begin{pmatrix} \omega & h_4 & h_5 \\ t_1 & \alpha_1 & \alpha_2 \\ t_2 & \alpha_3 & \alpha_4 \\ t_3 & \alpha_5 & \alpha_6 \end{pmatrix}, \begin{pmatrix} \omega\,(1 + \alpha_1 + \alpha_3) & h_4 & h_5 \\ t_1 & \frac{(\alpha_1+\alpha_3)\,(1+\alpha_1+\alpha_3+\alpha_5)}{1+\alpha_1+\alpha_3} & \frac{(\alpha_2+\alpha_4)\,(1+\alpha_1+\alpha_3+\alpha_5)}{1+\alpha_1+\alpha_3} \\ t_3 & \frac{\alpha_5}{1+\alpha_1+\alpha_3} & \frac{-\alpha_2\,\alpha_5-\alpha_4\,\alpha_5+\alpha_6+\alpha_1\,\alpha_6+\alpha_3\,\alpha_6}{1+\alpha_1+\alpha_3} \end{pmatrix}, \text{True}\right\}$$

hht

```
{β = B[ω, Sum[α_{3 i+j-5} t_i h_j, {i, 1, 2}, {j, 3, 5}]],
   O_1 = β // hm_{34→3} // sw_{13} // βCollect,
   O_2 = β // sw_{13} // sw_{14} // hm_{34→3} // βCollect;
   O_1 == O_2
  } /. α_i_ :→ î // ColumnForm
```

hht

$$\begin{pmatrix} \omega & h_3 & h_4 & h_5 \\ t_1 & \hat{1} & \hat{2} & \hat{3} \\ t_2 & \hat{4} & \hat{5} & \hat{6} \end{pmatrix}$$

$$\begin{pmatrix} \omega\left(1+\hat{1}+\hat{2}+\hat{1}\,\hat{2}+\hat{2}\,\hat{4}\right) & h_3 & h_5 \\ t_1 & \frac{(1+\hat{1}+\hat{4})\,(\hat{1}+\hat{2}+\hat{1}\,\hat{2}+\hat{2}\,\hat{4})\,(1+\hat{2}+\hat{5})}{1+\hat{1}+\hat{2}+\hat{1}\,\hat{2}+\hat{2}\,\hat{4}} & \frac{\hat{3}\,(1+\hat{1}+\hat{4})\,(1+\hat{2}+\hat{5})}{1+\hat{1}+\hat{2}+\hat{1}\,\hat{2}+\hat{2}\,\hat{4}} \\ t_2 & \frac{\hat{4}+\hat{5}+\hat{1}\,\hat{5}+\hat{4}\,\hat{5}}{1+\hat{1}+\hat{2}+\hat{1}\,\hat{2}+\hat{2}\,\hat{4}} & \frac{-\hat{3}\,\hat{4}-\hat{3}\,\hat{5}-\hat{1}\,\hat{3}\,\hat{5}-\hat{3}\,\hat{4}\,\hat{5}+\hat{6}+\hat{1}\,\hat{6}+\hat{2}\,\hat{6}+\hat{1}\,\hat{2}\,\hat{6}+\hat{2}\,\hat{4}\,\hat{6}}{1+\hat{1}+\hat{2}+\hat{1}\,\hat{2}+\hat{2}\,\hat{4}} \end{pmatrix}$$

True

R3

```
{(R^-)_{51} (R^-)_{62} (R^+)_{34} // gm_{14→1} // gm_{25→2} // gm_{36→3},
  (R^+)_{61} (R^-)_{24} (R^-)_{35} // gm_{14→1} // gm_{25→2} // gm_{36→3}}
```

R3

$$\left\{\begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+T_2}{T_2} & 0 \\ t_3 & \frac{-1+T_3}{T_2} & -\frac{-1+T_3}{T_3} \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+T_2}{T_2} & 0 \\ t_3 & \frac{-1+T_3}{T_2} & -\frac{-1+T_3}{T_3} \end{pmatrix}\right\}$$

8_17-1

```
β = (R^-)_{12,1} (R^-)_{27} (R^-)_{83} (R^-)_{4,11} (R^+)_{16,5} (R^+)_{6,13} (R^+)_{14,9} (R^+)_{10,15}
```

8_17-1

1	h_1	h_3	h_5	h_7	h_9	h_{11}	h_{13}	h_{15}
t_2	0	0	0	$-\frac{-1+T_2}{T_2}$	0	0	0	0
t_4	0	0	0	0	0	$-\frac{-1+T_4}{T_4}$	0	0
t_6	0	0	0	0	0	0	$-1+T_6$	0
t_8	0	$-\frac{-1+T_8}{T_8}$	0	0	0	0	0	0
t_{10}	0	0	0	0	0	0	0	$-1+T_{10}$
t_{12}	$-\frac{-1+T_{12}}{T_{12}}$	0	0	0	0	0	0	0
t_{14}	0	0	0	0	$-1+T_{14}$	0	0	0
t_{16}	0	0	$-1+T_{16}$	0	0	0	0	0

8_17-2

Do[β = β // gm$_{1k\to 1}$, {k, 2, 10}]; β

8_17-2

$$\begin{pmatrix} \frac{T_1^2+T_{16}-T_1 T_{16}}{T_1^2} & h_1 & h_{11} & h_{13} & h_{15} \\ t_1 & -\frac{(-1+T_1) T_{14} (T_1^3+T_{16}^2)}{T_1^2 T_{12} (T_1^2+T_{16}-T_1 T_{16})} & -\frac{(-1+T_1)(1-T_1+T_1^2) T_{14} T_{16}}{T_1 (T_1^2+T_{16}-T_1 T_{16})} & \frac{(-1+T_1)(1-T_1+T_1^2) T_{14}}{T_1^2+T_{16}-T_1 T_{16}} & -1+T_1 \\ t_{12} & -\frac{-1+T_{12}}{T_{12}} & 0 & 0 & 0 \\ t_{14} & \frac{(-1+T_{14})(-T_1+T_1^2+T_{16})}{T_{12} (T_1^2+T_{16}-T_1 T_{16})} & \frac{(-1+T_1)(1-T_1+T_1^2)(-1+T_{14}) T_{16}}{T_1 (T_1^2+T_{16}-T_1 T_{16})} & -\frac{(-1+T_1)(1-T_1+T_1^2)(-1+T_{14})}{T_1^2+T_{16}-T_1 T_{16}} & 0 \\ t_{16} & \frac{T_1 (-1+T_{16})}{T_{12} (T_1^2+T_{16}-T_1 T_{16})} & \frac{(-1+T_1) T_1 (-1+T_{16})}{T_1^2+T_{16}-T_1 T_{16}} & -\frac{(-1+T_1)^2 (-1+T_{16})}{T_1^2+T_{16}-T_1 T_{16}} & 0 \end{pmatrix}$$

8_17-3

Do[β = β // gm$_{1k\to 1}$, {k, 11, 16}]; β

8_17-3

$$\begin{pmatrix} -\frac{1-4 T_1+8 T_1^2-11 T_1^3+8 T_1^4-4 T_1^5+T_1^6}{T_1^3} \\ t_1 \end{pmatrix}$$

8_17-4

```
Alexander[Knot[8, 17]][X]
```

8_17-4

KnotTheory::loading : Loading precomputed data in PD4Knots`.

8_17-4

$$11-\frac{1}{X^3}+\frac{4}{X^2}-\frac{8}{X}-8 X+4 X^2-X^3$$

Recycling

StandardAlexander

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & T-1 & 0 & -T \\ -1 & T & 0 & 0 & 0 & 0 & 1-T & 0 \\ 0 & -1 & T & 0 & 1-T & 0 & 0 & 0 \\ T-1 & 0 & -T & 1 & 0 & 0 & 0 & 0 \\ 0 & 1-T & 0 & -1 & T & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -T & 1 & 0 & T-1 \\ 0 & 0 & 1-T & 0 & 0 & -1 & T & 0 \\ 0 & 0 & 0 & T-1 & 0 & 0 & -T & 1 \end{pmatrix}$$ **[[1 ;; 7, 1 ;; 7]] // Det**

StandardAlexander

$$-1+4 T-8 T^2+11 T^3-8 T^4+4 T^5-T^6$$

Work in Progress

```
GD[K_] := GD @@ (
   PD[K] /. X[i_, j_, k_, l_] :→ If[PositiveQ[X[i, j, k, l]],
      Ar[l, i, +1], Ar[j, i, -1]
     ]
  )
```

```
βZ[L_] := Module[
  {skel, β, s, k},
  skel = Skeleton[L];
  β = Times @@ GD[L] /. {Ar[x_, y_, +1] :> (R⁺)_xy, Ar[x_, y_, -1] :> (R⁻)_xy};
  Do[
   Do[
    β = β // gm_skel[[s,1]], skel[[s,k]]→skel[[s,1]],
    {k, 2, Length[skel[[s]]]}
   ],
   {s, Length[skel]}
  ];
  β
 ]
```

```
βZ[Knot[8, 17]][[1]]
```

$$-\frac{1 - 4 T_1 + 8 T_1^2 - 11 T_1^3 + 8 T_1^4 - 4 T_1^5 + T_1^6}{T_1^2}$$

```
Factor[βZ[#][[1]] / Alexander[#][T₁]] & /@ AllKnots[{3, 8}]
```

$$\left\{\frac{1}{T_1}, T_1, \frac{1}{T_1^2}, \frac{1}{T_1^2}, 1, 1, 1, \frac{1}{T_1^3}, \frac{1}{T_1^3}, T_1^4, T_1^4, \frac{1}{T_1^3}, \frac{1}{T_1}, T_1^2, \frac{1}{T_1}, \frac{1}{T_1}, T_1, T_1, T_1^3, \frac{1}{T_1}, T_1, T_1, T_1, T_1, \frac{1}{T_1}, T_1, T_1, \frac{1}{T_1}, \frac{1}{T_1^3}, \frac{1}{T_1}, T_1, 1, T_1^4, 1, \frac{1}{T_1}\right\}$$

```
βCollect[Bu[ω_, λ_, μ_]] := Bu[
   βSimp[ω],
   Collect[λ, h_, βSimp],
   Collect[μ, h_, Collect[#, t_, βSimp] &]
  ];
Bu[ηs_List, B[ω_, μ_]] := Module[{λ},
   λ = (1 + Coefficient[μ, #] /. t_ → 1) & /@ ηs;
   Bu[ω,
     Thread[ηs → λ],
     -μ + (ηs /. h_a_ :> t_a h_a).λ
    ] // βCollect
  ];
B[Bu[ω_, λ_, μ_]] := 0;
```

```
β0 = βZ[L = Link["L6a5"]]
```

KnotTheory::loading : Loading precomputed data in PD4Links`.

$$\begin{pmatrix} \frac{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)}{T_1^2 T_5^2 T_9^2} & h_1 & h_5 \\ t_1 & -\frac{(-1+T_1)(1-T_1-T_5-T_9+T_5 T_9+T_1 T_5 T_9)}{T_5(-1+T_1+T_5) T_9 (-1+T_1+T_9)} & -\frac{(-1+T_1) T_1}{(-1+T_1+T_5)(-1+T} \\ t_5 & -\frac{(-1+T_5)(-T_1-T_5+T_1 T_5+T_1 T_9+T_5 T_9)}{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)} & -\frac{(-1+T_5)(-1+2 T_1-T_1^2+T_5-T_1 T_5+2 T_9-2 T_1 T_9-T_5 T_9}{T_1(-1+T_1+T_5) T_9 (-1+T_1+T} \\ t_9 & -\frac{(-1+T_9)(1-T_1-T_5+T_1 T_5-T_9+T_1 T_9+T_5 T_9)}{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)} & -\frac{(-1+T_9)(-T_5+T_1 T_5-T_9+}{(-1+T_1+T_5)(-1+T_1+T_9)} \end{pmatrix}$$

```
Bu[{h1, h5, h9}, β0]
```

$$\text{Bu}\Big[\frac{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)}{T_1^2 T_5^2 T_9^2}, \left\{h_1 \to \frac{1}{T_5 T_9}, h_5 \to \frac{1}{T_1 T_9}, h_9 \to \frac{1}{T_1 T_5}\right\},$$
$$h_9 \left(\frac{t_1(-1+T_1)}{-1+T_1+T_9} + \frac{t_5(-1+T_5) T_9}{(-1+T_1+T_9)(-1+T_5+T_9)} + \frac{t_9 T_9^2}{(-1+T_1+T_9)(-1+T_5+T_9)}\right) +$$
$$h_1 \left(\frac{t_1 T_1^2}{(-1+T_1+T_5)(-1+T_1+T_9)} + \frac{t_5(-1+T_5)(-T_1-T_5+T_1 T_5+T_1 T_9+T_5 T_9)}{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)} + \frac{t_9(-1+T_9)(1-T_1-T_5+T_1 T_5-T_9+T_1 T_9+T_5 T_9)}{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)}\right) +$$
$$h_5 \left(\frac{t_1(-1+T_1) T_1}{(-1+T_1+T_5)(-1+T_1+T_9)} + \frac{t_9(-1+T_9)(-T_5+T_1 T_5-T_9+T_1 T_9+T_5 T_9)}{(-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)} + \left(t_5\left(-1+T_1+T_5-T_1 T_5-T_5^2+T_1 T_5^2+T_9-T_1 T_9-T_5 T_9+T_1 T_5 T_9+T_5^2 T_9\right)\right) \Big/ \left((-1+T_1+T_5)(-1+T_1+T_9)(-1+T_5+T_9)\right)\right)\Big]$$

```
βMVA[Bu[ω_, λ_, μ_]] := Module[
   {lbls, mat},
   lbls = Rest[First /@ λ];
   mat = Outer[
     Coefficient[μ - lbls.(lbls /. h_a_ :> t_a), #1 * #2] &,
     lbls, lbls /. h_a_ :> t_a
    ];
   ω * Det[mat] / (1 - λ[[1, 1]] /. h_i_ → T_i) // Factor
  ];
βMVA[L_Link] := βMVA[Bu[h_# & /@ (First /@ Skeleton[L]), βZ[L]]]
```

```
βMVA[L]
```

$$-\frac{-T_1-T_5+T_1 T_5-T_9+T_1 T_9+T_5 T_9}{T_1^2 T_5^2 T_9^2}$$

```
βMVA[L = Link["L8a16"]]
```

$$-\frac{(-1+T_1)(-1+T_5)(-1+T_{11})(1+T_5 T_{11})}{T_1 T_5 T_{11}}$$

```
βZ[L]
```

$$\left(\frac{1-2 T_1+T_1^2-2 T_5+4 T_1 T_5-2 T_1^2 T_5+T_5^2-2 T_1 T_5^2+T_1^2 T_5^2-2 T_{11}+4 T_1 T_{11}-2 T_1^2 T_{11}+4 T_5 T_{11}-10 T_1 T_5 T_{11}+6 T_1^2 T_5 T_{11}-2 T_5^2 T_{11}+8 T_1 T_5^2 T_{11}-5 T_1^2 T_5^2 T_{11}-\cdots}{}\right.$$

```
Simplify[1/βMVA[#] (MultivariableAlexander[#][T] /. T[i_] :> T_Skeleton[#][[i,1]])] & /@
  AllLinks[8]
```

KnotTheory::loading : Loading precomputed data in MultivariableAlexander4Links`.

$$\Big\{-\sqrt{T_1}\, T_5^{3/2},\ -\frac{\sqrt{T_1}}{\sqrt{T_5}},\ -T_1^{3/2} T_5^{3/2},\ -\sqrt{T_1}\, T_5^{3/2},\ -\frac{T_1^{3/2}}{\sqrt{T_5}},\ -\frac{T_1^{3/2}}{\sqrt{T_5}},\ -T_1^{3/2} T_5^{7/2},\ -\frac{T_1}{T_7},\ -T_1 T_7,$$
$$-T_1^2 T_7^3,\ -T_1^2 T_7^3,\ -T_1^{5/2} T_9^{5/2},\ -T_1^{5/2} T_9^{5/2},\ -T_1^{5/2} T_9^{5/2},\ -T_1^{3/2} T_5^{3/2} \sqrt{T_9},\ -\sqrt{T_1},\ -T_1^{3/2} T_5^2 T_{11}^2,$$
$$-\frac{\sqrt{T_1}}{T_{11}^2},\ -\frac{\sqrt{T_1}\, T_5}{T_{11}},\ -\frac{T_1^{3/2}\sqrt{T_5}}{T_{13}^{3/2}},\ -T_1^{3/2} T_5^{3/2} T_9^{3/2} T_{13}^{3/2},\ -T_1^{3/2} T_5^{3/2},\ -\sqrt{T_1}\sqrt{T_5},\ -T_1^{3/2} T_5^2 T_{11}^2,$$
$$-\sqrt{T_1}\, T_5^2 T_{11},\ -\sqrt{T_1}\, T_5^{3/2} T_{11}^{3/2},\ -T_1^{3/2} T_5^{5/2}\sqrt{T_{13}},\ -\frac{\sqrt{T_1}\sqrt{T_5}}{\sqrt{T_9}\sqrt{T_{13}}},\ -\sqrt{T_1}\sqrt{T_5}\sqrt{T_9}\sqrt{T_{13}}\Big\}$$